# Maintainer: Filipe Laíns (FFY00) <lains@archlinux.org>

_pkgname=libusb1
pkgname=python-$_pkgname
pkgver=3.0.0
pkgrel=2
pkgdesc='Python ctype-based wrapper around libusb1'
arch=('any')
url='https://github.com/vpelletier/python-libusb1'
license=('LGPL2.1')
depends=('python' 'libusb')
makedepends=('python-build' 'python-installer' 'python-setuptools' 'python-wheel')
source=("$pkgname-$pkgver.tar.gz::$url/archive/$pkgver.tar.gz")
sha512sums=('7812dc4a22fc277f9791fb7050aa0a9c3953d1d3b386bb848380026a036a09bdff7745163eab07f5b70f919cb45ea4cd14f212f9e065dee670afee97a458de9d')

build() {
  cd $pkgname-$pkgver

  python -m build -nw
}

package() {
  cd $pkgname-$pkgver

  python -m installer -d "$pkgdir" dist/*.whl
}

# vim:set ts=2 sw=2 et:
